
1 

Paper 8780-2016  

File Management Using Pipes and X Commands in SAS® 
Emily K.Q. Sisson, Boston University School of Public Health 

ABSTRACT  
SAS for Windows can be an extremely powerful piece of software, not only for analyzing data, but also for 
organizing and maintaining output and permanent datasets.  By employing pipes and operating system 
(‘X’) commands within a SAS session, you can easily and effectively manage files of all types stored on 
your local network.   

INTRODUCTION  
Not only are SAS programmers responsible for creating, maintaining, and documenting SAS programs, 
but also the associated output and permanent datasets. SAS can be a very handy tool in managing the 
various files occupying network folders. 

While this paper will display syntax specific to file management on a Windows-based computer, the 
concepts extend to UNIX- and Linux-based systems. 

INVENTORYING YOUR NETWORK 

GENERATE A FILE LISTING  

The first step in understanding what files need to be managed is to import a directory listing of file 
attributes into a SAS dataset.  Using a pipe device type on a filename statement allows you to invoke 
DOS commands within a SAS session.  By executing the DOS command for a directory listing (“dir”), an 
inventory of the specified directory can be input into a SAS dataset.  Additionally, the “/s” option should be 
included in the DOS command to recursively process any subfolders within that directory.  Since the 
length of each record in the directory will differ, it is prudent to input the data into a SAS dataset using the 
$VARYINGw. format.  The LENGTH option will assign the internally stored record length to RECLEN, 
which then becomes the length-variable argument to the $VARYINGw. format.  An example of this code 
is shown here: 

filename dirlist pipe 'dir "Y:\MyFiles" /s'; 
data dirlist ; 
     length lineinfo $256 ; 
     infile dirlist length=reclen ; 
     input lineinfo $varying256. reclen ; 
run; 
 

As seen in Display 1, the resulting DIRLIST dataset is one that contains many observations, but only one 
column.  All the information gathered through the DOS “dir” command is stored in a text string.   

 
Display 1. Excerpt from the DIRLIST dataset 

MANIPULATE A FILE LISTING  

Once there exists a dataset containing information about directory locations and file attributes, it can be 
used to manage files.  For illustrative purposes, this paper will consider a scenario where you need to 
create archive folders within the existing directory and subsequently move files to each based on file date 
and type.  Within the root folder, the intent is to have one subfolder per year and within each subfolder 
have a folder for each file type (PDF, Word Document, etc.). 

Fortunately, the text string stored in the DIRLIST dataset can be manipulated, truncated, and excerpted to 
suit a variety of needs.  For example, you can create variables representing file path, file name, file 
extension, and date of last modification.  With some simple code, this one column dataset can be 
morphed into a more useful tool for managing files on a network:   

data dirlist_useful; 
 set dirlist; 
 /*Path of the directory appears once, use retain statement to assign*/ 
 length directory $1000; 
 retain directory; 
 if left(upcase(lineinfo))=: 'DIRECTORY OF' then  

directory = substr(left(lineinfo),14); 
 
 /*Isolate other important information*/ 
 filename = substr(left(lineinfo),40);  
 fileextens = scan(strip(lineinfo),-1);  
 filedate = input(substr(lineinfo,1,10),?? mmddyy10.);  
  format filedate mmddyy10.; 
 filetime = input(scan(lineinfo,4)||" "||scan(lineinfo,5),time12.);  
  format filetime time12.; 
 
 /*Categorize file extensions as file types, extract year from date*/ 
   [lines omitted] 
 
 /*Delete extraneous rows*/ 
 if lineinfo = '' then delete; 
 if index(upcase(lineinfo),'<DIR>') then delete; 
 if left(upcase(lineinfo)) =: 'VOLUME' then delete; 
 if left(upcase(lineinfo))=: 'DIRECTORY OF' then delete; 
 if fileextens in ('bytes' 'Listed:' 'free') then delete; 
run; 
 

Display 2 shows a selection of the newly created dataset.  This will prove significantly more useful than 
the previous iteration. 

 
Display 2. Excerpt from the DIRLIST_USEFUL dataset 

MANAGING YOUR NETWORK 

CREATE NEW NETWORK FOLDERS 

In addition to pipe devices, SAS can communicate with a network through operating system (‘X’) 
commands.  When an ‘X’ command is invoked within SAS, you are placed in a Windows command 
prompt session, and any subsequent statements are executed as Windows commands.  Among other 
things, these commands can be used to make or delete directories, and move, copy, and delete individual 
files.  The ‘X’ command mkdir can be used to create a new folder in a directory. The –p option (“parent”) 
will create any intermediate directories should they not already exist: 

x "mkdir –p Y:\MyFiles\newfolder"; 
 

Once in this command prompt session, you must type ‘exit’ into the command prompt window before 
returning to the SAS session.  Luckily, there is a system option in SAS to avoid having to type ‘exit’ after 
each command: noxwait. Since this paper is focusing on archiving by both file type and date, using a 
macro to invoke the mkdir command repeatedly is the most efficient means to create your desired 
subfolders: 

/*Isolate unique combos of Directory/Year/File Type to create folders*/ 
proc sql;   
 create table dir_yr_typ as  
 select distinct directory, year, filetype  
 from dirlist_useful; 
quit; 
 
/*Using the dir_yr_typ dataset, create a folder for every combo of yr/typ*/ 
%macro createdir(dir=,yr=,typ=); 
 x "mkdir -p  &dir.\&yr.\&typ."; 
%mend createdir; 
 
options noxwait; 
data _null_; 
 set dir_yr_typ; 
 command = cats('%createdir(dir=',directory,', yr=',year,',  

typ=',filetype,');'); 
 call execute(command); 
run; 
 

After isolating unique combinations of directory/year/file type, the call execute function is used within a 
data _NULL_ step to invoke the %createdir macro many times over.  Display 3 illustrates the newly 
defined “command” column in the data _NULL_ step.  Through the use of the call execute function, the 
macro is invoked once per observation by processing the macro call stored in “command”. 

 
 Display 3. Excerpt from the DATA _NULL_ step 

REORGANIZE FILES 

Similarly, a simple macro can be created to move existing files to their new destinations.  Here, the ‘X’ 
command of choice is the move command followed by a file’s current location then new location.  The 
DIRLIST_USEFUL dataset is used in a data _NULL_ step to generate a new set of macro invocations, 
this time for the newly defined %movfil macro: 

/*Using the original dirlist_useful set, move old files to new 
destinations*/ 
%macro movfil(dir=,name=,yr=,typ=); 
 x move "&dir\&name" "&dir\&yr\&typ\&name"; 
%mend movfil; 
 
data _null_; 
 set dirlist_useful; 
 command = cats('%movfil(dir=',directory,', name=',filename,',  

typ=',upcase(filetype),',yr=',year,');'); 
 call execute(command); 
run; 
 

Once the DATA _NULL_ step has been executed, the hundreds of files found in the original folder have 
been moved to their new locations. Having used pipes to inventory your directory and ‘X’ commands to 
subsequently restructure it, your documents are much more organized.  Displays 4a and 4b demonstrate 
the original and final layouts of the directory. 

 

 
Display 4a. Original Layout of Y:\MyFiles 

 
Display 4b. Restructured Layout of Y:\MyFiles 

CONCLUSION 
This paper highlights the value of being able to execute Windows operations within a SAS session.  Using 
pipes and operating system commands, you can inventory, manage, and create a complex hierarchy of 
varying file types very easily.  With thoughtful use of the call execute function, hundreds, if not thousands, 
of files can be manipulated in a matter of seconds.  Archiving files is just one of many applications 
possible with these tools. 

